

Practical assignment №4

3D PROBLEM OF CONTACT BETWEEN TWO DEFORMABLE SOLIDS

Keywords

1. Theory of elasticity – structural analysis
2. 3D problem
3. Static problem
4. Contact
5. Nonlinear analysis

PROBLEM DESCRIPTION

Let us consider a problem of contact between two deformable solids with different material properties. We need to calculate contact stresses and compare the results with theoretical values obtained by formulas of the Hertz theory.

Problem statement.

We consider a problem of interaction between two rollers with different physical properties, geometrical parameters and radii of the notches in two perpendicular directions (Fig. 1).




Fig. 1 Model sketch

The scheme of the model in the edge plane of the rollers is shown in Fig. 2. The upper roller with the notch curvature of the radius $R_3 = 0.012$ (m) has the following dimensions: radius $R_1 = 0.02$ (m) and width $2h_1$; $h_1 = 0.004$ (m). The lower roller with the notch curvature of the radius $R_4 = 0.015$ (m) and the notch height $h_3 = 0.0005$ (m) has the radius $R_2 = 0.02$ (m) and the width $2h_2$; $h_2 = 0.005$ (m).

The rollers have different physical properties. The upper roller is made of an isotropic material with the Young's modulus $E_1 = 2.5 \cdot 10^{11}$ (N/m²) and the

Poisson's ratio $\nu_1 = 0.29$, the lower roller is made of an isotropic material with the Young's modulus $E_2 = 2 \cdot 10^{11}$ (N/m²) and the Poisson's ratio $\nu_2 = 0.29$.

The contact interaction between the rollers is induced by the concentrated force $P = 225$ (N), applied to the center of the upper roller and pointed downwards. The horizontal axis of the lower roller is rigidly fixed. It is necessary to determine the stresses σ_y and the displacements u_y in the contact zone between two rollers.




Fig. 2 Projection in the edge plane

SOLVING THE PROBLEM USING ANSYS

Input file for this problem is St3SCONT.INP.

We can use the properties of symmetry in order to build a solid model in ANSYS. We can note that the geometrical model shown in Fig. 1 has two planes of symmetry, OXY and OYZ. Therefore, we can build a 1/4 part of the original domain (Fig. 3) and then apply the symmetry properties to the areas lying on the symmetry planes.



Fig. 3 A quarter of the domain

We begin with the command **/TITLE** to specify the title of the problem. This title will be shown in ANSYS window:

```
/TITLE, Contact interaction of a roller pair
```

In the preprocessor **/PREP7** we define the parameters for the values of the geometrical sizes, material properties, finite element sizes and load magnitudes. For the finite element mesh we will use the mesh condensation in the contact zone. The parameters for the finite element mesh can be taken as follows. We introduce the characteristic finite element size **GLSIZE** for the finite element mesh outside the contact area. In auxiliary volumes around the contact zone we will use smaller size **SIZE_ALL**. Yet another size **SIZE_0** will be used to condense the finite element mesh around the contact point.

```
/PREP7
! Geometrical sizes for the upper and lower rollers
R1=0.02      ! Radius of the upper roller
R2=0.02      ! Radius of the lower roller
R3=0.012     ! Radius of the notch curvature for the upper roller
R4=0.015     ! Radius of the notch curvature for the lower roller
h1=0.004     ! Half width of the upper roller
h2=0.005     ! Half width of the lower roller
h3=0.0005    ! Notch depth of the lower roller

! Material constants for the upper and lower rollers
E1=2.5e11    ! Young's modulus of the upper roller
E2=2e11      ! Young's modulus of the lower roller
NU1=0.29     ! Poisson's ratio of the upper roller
NU2=0.29     ! Poisson's ratio of the lower roller

! Geometrical sizes of the initial areas for the upper and lower rollers
d1=0.003     ! width of the contact zone on the upper roller
d2=0.003     ! width of the contact zone on the lower roller
PSIX_BOTTOM=12 ! Opening angle along X-axis for the lower initial area
PSIX_TOP=15  ! Opening angle along X-axis for the upper initial area
PSIZ_BOTTOM=21 ! Opening angle along Z-axis for the lower initial area
PSIZ_TOP=18  ! Opening angle along Z-axis for the upper initial area

! Meshing parameters
GLSIZE=0.0029 ! Characteristic element size
SIZE_ALL=GLSIZE/5 ! Element size for the auxiliary volumes around the contact zone
SIZE_0=GLSIZE/20 ! Element size around the contact point

!Force value
PY=0.225e3
```

Next we will set the material properties. For a static analysis of a linear isotropic material we need to define the Young's modulus E and the Poisson's

ratio ν . In this problem we have two different materials, so we need to define two sets of material properties.

```
! Define material properties
MP,EX,1,E1      ! Young's modulus of the upper roller
MP,NUXY,1,NU1  ! Poisson's ratio of the upper roller
MP,EX,2,E2      ! Young's modulus of the lower roller
MP,NUXY,2,NU2  ! Poisson's ratio of the lower roller
```

We solve the problem in a 3D setting. In order to mesh the volumes of the rollers we will use 3D 10-node tetrahedral finite elements SOLID92, which have three degrees of freedom u_x , u_y , u_z at each node. On the contact surfaces we should set contact and target elements (such called contact pairs). In order to simulate the contact on the surface of the lower roller, we will use 8-node surface-to-surface contact elements CONTA174, and for the contact zone of the upper roller we will use target elements TARGE170.

```
! Element types
ET,1,SOLID92 ! 3-D quadratic 10-node element
ET,2,TARGE170 ! 3-D contact element
ET,3,CONTA174 ! 3-D target element
```

In order to create a solid model, we will first build 2D initial areas and then rotate them around the axis at the given angle using VROTAT command. It makes sense to build small auxiliary volumes around the contact zone to ensure mesh condensation in the vicinity of the contact point.

```
! Solid model
! Define keypoints
K,1,0,0
K,2,sqrt(h3*(2*R4-h3)),h3
K,3,h2,h3
K,4,h2,-(R2-h3)
K,5,0,-(R2-h3)
K,6,0,R3
K,7,0,R4
K,8,0,R1
K,9,h1,R1
K,10,h1,R3-sqrt(R3*R3-h1*h1)
K,11,0,0
K,12,0,0
K,13,0,R4
CLOCAL,11,1,0,R4,0 ! Define cylindrical coordinate system 11
CSYS,11 ! Change actual CS to a user-defined CS 11
K,14,R4,-90+PSIX_BOTTOM
K,15,R4+d2,-90+PSIX_BOTTOM
K,16,R4+d2,-90
CSYS,0 ! Change actual CS to Global Cartesian
K,17,0,R3
CLOCAL,22,1,0,R3,0 ! Define cylindrical coordinate system 22
```

```

CSYS,22 ! Change actual CS to a user-defined CS 22
K,18,R3,-90
K,19,R3,-90+PSIX_TOP
K,20,R3-d1,-90+PSIX_TOP
K,21,R3-d1,-90
CSYS,0 ! Change actual CS to Global Cartesian

! Define arcs 1-6
LARC,1,2,7,R4
LARC,11,10,6,R3
LARC,12,14,13,R4
LARC,16,15,13,R4+d2
LARC,18,19,17,R3
LARC,21,20,17,R3-d1

! Define straight lines 7-17
L,12,16
L,14,15
L,21,18
L,20,19
L,2,3
L,3,4
L,4,5
L,5,1
L,11,8
L,8,9
L,9,10

! Define areas 1-4 by keypoints
A,11,8,9,10
A,1,2,3,4,5
A,21,20,19,18
A,12,14,15,16

! Create a volume by rotating area 3 around the axis with
keypoints 8 and 9 at the angle PSIZ_BOTTOM
VROTAT,3,,,,,8,9,PSIZ_BOTTOM
! Create a volume by rotating area 4 around the axis with
keypoints 4 and 5 at the angle PSIZ_TOP
VROTAT,4,,,,,4,5,PSIZ_TOP
! Create a volume by rotating area 1 around the axis with
keypoints 8 and 9 at the angle of 180 degrees
VROTAT,1,,,,,8,9,180
! Create a volume by rotating area 2 around the axis with
keypoints 4 and 5 at the angle of 180 degrees
VROTAT,2,,,,,4,5,180

! Overlap volumes (generate new volumes in place of intersecting
volumes)
VOVLAP,ALL

```

Now we will associate the sets of material properties with the volumes of the upper and lower rollers:

```

! Associate sets of material properties with volumes
VSEL,S,VOLU,,1 ! Select volume 1
VSEL,A,VOLU,,4 ! Add volume 4 to the set
VSEL,A,VOLU,,8 ! Add volume 8 to the set
VATT,1,,1 ! Associate material and element type attributes
MAT=1, TYPE=1 with selected volumes
VSEL,S,VOLU,,9 ! Select volume 9
VSEL,A,VOLU,,6,7! Add volumes 6 and 7 to the set
VATT,2,,1! Associate material and element type attributes MAT=2,
TYPE=1 with selected volumes
VSEL,ALL ! Select all volumes

```

The finite element size around main keypoints is defined by **KESIZE** command. Here we define three sizes of finite elements. The first size is set for auxiliary volumes around the contact zone. The second size ensures the mesh condensation around the contact point. The third size is used to mesh four volumes outside of the contact zone.

```

! Set the first element size around all keypoints
KESIZE,ALL,SIZE_ALL
! Set the second element size in the vicinity of the contact
point
KESIZE,18,SIZE_0
! Mesh volumes 1 and 7
VMESH,1
VMESH,7
! Set the third element size around all keypoints
KESIZE,ALL,GLSIZE
! Mesh volumes 4,6,8 and 9
VMESH,4
VMESH,6
VMESH,8
VMESH,9

```

The resulting finite element mesh is shown in Fig. 4.




Fig. 4 Mesh condensation around the contact zone

Next we will generate contact pairs, which will consist of contact and target elements. The contact elements will be on the surface of the lower roller, and the target elements will be on the surface of the upper roller. The contact and target elements can be created by ESURF command which generates elements on the free faces of the existing elements. The target and contact elements will be generated on the contact surfaces which lie on the areas of auxiliary volumes.

```

! Define contact elements
ASEL,S,AREA,,34 ! Select area 34
ASEL,A,AREA,,40 ! Add area 40
NSLA,,1 ! Select nodes associated with selected areas
CM,STRIP,NODE ! Combine selected nodes into a group STRIP
TYPE,3 ! Set element type 3 (contact)
MAT,2 ! Set material type 2
ESURF ! Generate contact elements for the selected nodes
ASEL,S,AREA,,7 ! Select area 7
ASEL,A,AREA,,38 ! Add area 38
NSLA,,1 ! Select nodes associated with selected areas
TSHAP,CIRC ! Set circular shape for target elements
CM,PUNCH,NODE ! Combine selected nodes into a group PUNCH
TYPE,2 ! Set element type 2 (target)
MAT,1 ! Set material type 1
ESURF ! Generate contact elements for the selected nodes
NSEL,ALL ! Select all nodes
ASEL,ALL ! Select all areas

SAVE
FINISH

```

Next stage is to set boundary conditions and solve the problem. These actions can be performed in solver **/SOLU**. The lower roller should be rigidly fixed at the line which goes through the center of the lower roller along OX-axis. We also need to set conditions of symmetry at the areas which lie in OXY and OYZ planes. In the center of the upper roller we will set the concentrated force with the value PY/4 directed along OY-axis (*here we should set a quarter of the force value because we consider a quarter of the original domain*).

```

/SOLU
! Lower roller is rigidly fixed
LSEL,S,LINE,,13 ! Select line 13
NSLL,,1 ! Select nodes associated with selected lines
D,ALL,UX,0 ! Define UX=0 at all selected nodes
D,ALL,UY,0 ! Define UY=0 at all selected nodes
D,ALL,UZ,0 ! Define UZ=0 at all selected nodes
NSEL,ALL ! Select all nodes
LSEL,ALL ! Select all lines

! Define symmetry boundary conditions
DA,3,SYMM ! Generate symmetry constraints on area 3

```

```

DA,8,SYMM ! Generate symmetry constraints on area 8
DA,19,SYMM ! Generate symmetry constraints on area 19
DA,22,SYMM ! Generate symmetry constraints on area 22
DA,31,SYMM ! Generate symmetry constraints on area 31
DA,32,SYMM ! Generate symmetry constraints on area 32
DA,33,SYMM ! Generate symmetry constraints on area 33
DA,35,SYMM ! Generate symmetry constraints on area 35
DA,36,SYMM ! Generate symmetry constraints on area 36
DA,37,SYMM ! Generate symmetry constraints on area 37
DA,39,SYMM ! Generate symmetry constraints on area 39
DA,41,SYMM ! Generate symmetry constraints on area 41

! Define force load
FK,8,FY,-PY/4 ! Define concentrated force at keypoint 8

SOLVE      ! Solve system of finite element equations
SAVE

FINISH

```

REVIEWING AND ANALYZING RESULTS

We will analyze the results in General Postprocessor. After several commands that set the options of the graphical output we issue **PLNSOL** command to plot pictures of the distribution of displacements **U_y** and stresses **S_y**:

```

/POST1
/DSCALE,ALL,OFF ! Turn off displacements scaling
/PLOPTS,INFO,1 ! Display legend on the right
PLNSOL,U,Y ! Plot displacements UY
! Use delay to view the previous picture
*ASK,TMP,ANY NUMBER OR PRESS "ENTER"
PLNSOL,S,Y ! Plot axial stresses SY

```

In order to verify that the displacements and stresses values for the upper and lower rollers are equal, we can compare the corresponding results.

In General postprocessor **POST1** using **ESEL** command we first select the elements of the upper roller and then using **NSLE** command we select the nodes associated with selected elements. Then we need to select the node located at the coordinate system origin. We can do this by implementing **NSEL** command and finding the number of this node by ***GET** command (the node number will be stored in parameter **N_CONT2**). Then we find the displacement **UY** at the node **N_CONT2**.

```

ESEL,S,MAT,,2 ! Select elements with attribute MAT=2
NSLE,,1 ! Select nodes associated with selected elements
! Select node located at the point with coordinates (0,0,0)
NSEL,S,LOC,X,0
NSEL,R,LOC,Y,0
NSEL,R,LOC,Z,0
! Set parameter N_CONT2 for the node with minimal number in this
location
*GET,N_CONT2,NODE,,NUM,MIN

```



```

NSEL,ALL ! Select all nodes
ESEL,ALL ! Select all elements
DELTA2=UY(N_CONT2)! Define displacement UY at node N_CONT2 as
DELTA2

```

Similarly we can find the displacement **UY** at the node located at the coordinate system origin for the lower roller (node **N_CONT1**).

```

ESEL,S,MAT,,1 ! Select elements with attribute MAT=1
NSLE,,1 ! Select nodes associated with selected elements
! Select node located at the point with coordinates (0,0,0)
NSEL,S,LOC,X,0
NSEL,R,LOC,Y,0
NSEL,R,LOC,Z,0
! Set parameter N_CONT1 for the node with minimal number in this
location
*GET,N_CONT1,NODE,,NUM,MIN
NSEL,ALL ! Select all nodes
ESEL,ALL ! Select all elements
DELTA1=UY(N_CONT1)! Define displacement UY at node N_CONT1 as
DELTA1

```

Using ***GET** command, we obtain the stresses **SY** at the nodes **N_CONT1** and **N_CONT2**.

```




! Define stress SY at node N_CONT1 as SYE_CONT1
*GET,SYE_CONT1,NODE,N_CONT1,S,Y
! Define stress SY at node N_CONT2 as SYE_CONT2
*GET,SYE_CONT2,NODE,N_CONT2,S,Y




```

Then in the scalar parameters listing we can check that the values of displacements and stresses at the contact point are equal for the upper and lower roller. This fact ensures accuracy of the results.

The computed values of the displacements u_y and the stresses σ_y at the contact point are the following: $u_y = -2,447 \cdot 10^{-6}$ (m), $\sigma_y = -1,214 \cdot 10^9$ (N/m²).

It is of interest to compare the computed values of stresses with theoretical values calculated by formulas of the Hertz theory. According to these formulas the maximal contact stresses equal $\sigma_y = 1,060 \cdot 10^9$ (N/m²). Thus, absolute values of the stresses computed in ANSYS are close to theoretical values.

№	Problem sketch	Geometrical sizes and boundary conditions
1		<p>Contact of elastic parallelepiped with the height $h=0.02$ m, width $w=0.01$ m and length $l=0.03$ m with elastic ball of the radius $R=0.01$ m. The bottom edge of the parallelepiped is rigidly fixed, and the ball is subjected to compression by the force $P=500$ N. The force is directed along the line which goes through the center of the ball and the point of their initial contact.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,57843 \sqrt[3]{\frac{P}{\eta^2 R^2}}$
2		<p>Contact of two elastic balls of the radii $R_1=0.01$ m (upper) and $R_2=0.02$ m (lower). The balls are compressed by the forces $P=700$ N. The forces are directed along the line which goes through the centers of the ball and the point of their initial contact.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,57843 \sqrt[3]{\frac{P}{\eta^2} \left(\frac{R_2 + R_1}{R_2 R_1} \right)^2}$
3	<p>Meridional section of two balls</p> 	<p>Contact of internal surface of elastic spherical layer of the radius $R_2=0.02$ m and thickness $h=0.005$ m with elastic ball of the radius $R_1=0.01$ m. Elastic spherical layer is rigidly fixed over the external surface. The ball inside it is subjected to the concentrated load $P=225$ N. The direction of the force goes through the centers of the balls and the point of initial contact.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,57843 \sqrt[3]{\frac{P}{\eta^2} \left(\frac{R_2 - R_1}{R_2 R_1} \right)^2}$

4	 <p>The diagram shows a 3D perspective of a gray cylindrical object resting on a gray rectangular block. The block is fixed to a surface, indicated by hatching. A series of downward-pointing arrows of varying lengths are distributed along the top surface of the cylinder, representing a distributed load. The label 'q' is placed above the arrows.</p>	<p>Contact of elastic parallelepiped with the width, height and length $w=0.01$ m, $h=0.02$ m and $l=0.03$ m, respectively, with elastic cylinder of the radius $R=0.01$ m. The lower surface of the parallelepiped is rigidly fixed, the cylinder is subjected to the distributed load of the value $q=75$ N/m per the unit length.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,5642 \sqrt{\frac{q}{\eta R}}$
5	 <p>The diagram shows two gray cylindrical objects in contact. The lower cylinder is fixed to a surface, indicated by hatching. The upper cylinder is positioned above it, touching at a single point. A single downward-pointing arrow labeled 'P' is shown at the point of contact, representing a concentrated load.</p>	<p>Contact of two elastic cylinders of the length $l=0.07$ m with mutually perpendicular axes and the radius $R=0.01$ m. The lower cylinder is rigidly fixed, and the upper cylinder is subjected to concentrated load $P=225$ N, directed downwards and going through the point of the initial contact of cylinders.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,57843 \sqrt{\frac{P}{\eta^2 R^2}}$
6	 <p>The diagram shows two gray cylindrical objects in contact. The lower cylinder is fixed to a surface, indicated by hatching. The upper cylinder is positioned above it, touching along a line. A series of downward-pointing arrows of varying lengths are distributed along the top surface of the upper cylinder, representing a distributed load. The label 'q' is placed above the arrows.</p>	<p>Contact of two elastic cylinders of the length $l=0.07$ m. The upper cylinder has the radius $R_1=0.01$ m and the lower cylinder has the radius $R_2=0.02$ m. The lower cylinder is rigidly fixed along the line. The upper cylinder is subjected to the distributed load of the value $q=40$ N/m per the unit length.</p> <p>The value of the maximal contact stress by the Hertz theory can be calculated by formula:</p> $\sigma_y = 0,5642 \sqrt{\frac{q R_2 + R_1}{\eta R_1 R_2}}$